
Adding 2D Mouse Support to UnrealScript 

Introduction 

This tutorial will guide you through the basics of adding a mode to your game that 
allows the mouse to be used to pick 2D objects on the screen. 
 
Please Note that it is the first step only allows you to pick 2D objects in the 
scene.  A subsequent tutorial expands this to 3D. 
 
By completing this tutorial you will have: 

• Adding a mode to switch between Normal controls and Mouse use 

• Display a texture to represent the cursor position 

• Recognise when the mouse is moved / clicked 
 
This will form the basics for an inventory system, decision, and different 
interaction mode such as your murder prediction. 
 
Once again this is not a full step-by-step guide.  You will need to alter some of 
this code to reflect the names of your classes and functions.  By now I’m 
assuming you’ll be able to spot these errors that the UnrealScript compiler flags. 

Add PlayerMousing state 

We need to switch the player into a completely different state.  This will mean 
that the mouse will no longer control the movement of the player in the world, but 
instead will control the mouse on the screen. 
 
These changes are made to your PlayerController class – mine is still 

named as ExampleController.uc 

 
I have included two functions.  One that is triggered when the player clicks the 
mouse and another that is triggered when the mouse is moved. 
 
You may need to add some variables at this point. E.g. PlayerMouse, 

ScreenV, LastHUDSizeX and LastHUDSizeY.  These are either type vector 

or float – I’m sure can spot which is which. 

 
 

state PlayerMousing 

{ 

   exec function Fire(float f) 

   { 

      // do stuff here for when players click their fire/select button   

 ExampleHud(MyHud).SetDisplayText("Fire"); 

      return; 

   } 

   simulated function PlayerMove(float DeltaTime) 

   { 

      local string textString; 

      local vector MouseV; 

 

      // get the new mouse position offset 

      MouseV.X= DeltaTime * aMouseX / 

(InputClass.default.MouseSensitivity * DesiredFOV * 0.01111); 

      MouseV.Y = DeltaTime * aMouseY / 

(InputClass.default.MouseSensitivity * DesiredFOV * -0.01111); 

 

      // update mouse position 

      PlayerMouse += MouseV; 

 

      // convert mouse position to screen coords, but only if we have 

good screen sizes 

      if ((LastHUDSizeX > 0) && (LastHUDSizeY > 0)) 

      { 

         ScreenV.X = PlayerMouse.X + LastHUDSizeX * 0.5; 

         ScreenV.Y = PlayerMouse.Y + LastHUDSizeY * 0.5; 

         // here is where you would use the screen coords to do a trace 

or check HUD elements  

   textString = ""$ScreenV.X$" "$ScreenV.Y; 

   ExampleHud(MyHud).SetDisplayText( textString ); 

      } 

 

      return; 

   } 

} 

 

Change State 

Our next step is to be able to switch states between normal and 
PlayerMousing. 

 
I added an exec function called MenuState() and have bound it to a 

keypress.  It stores the current state we are in and switches to our 
PlayerMousing state. 

 
You’ll need to add a variable to hold the ReturnState – it is of type name. 

 
 

exec function MenuState() 

{ 

 ReturnState = GetStateName(); 

 GotoState( 'PlayerMousing' ); 

} 

 

 
The next step is to be able to switch back to our previous state.  I have an 
identical exec function that is added to the PlayerMousing state – it simply 

changes our state back to our ReturnState. 

exec function MenuState() 

{   

 GotoState( ReturnState ); 

} 

 

Displaying the Cursor 

These changes are all made in your HUD class. 

 
Firstly, add a new texture variable called MouseCursorTexture.   

 
Ensure that you assign it in the defaultproperties section.  I used 

'2K4Menus.Cursors.Pointer' but you can hunt through the TextureBrowser 

for your own cursor – or add your own if you prefer. 
 
Next add a function to display the cursor.  This also limits where the mouse can 
move.  Again this makes heavy reference to my Controller class – you’ll need 

to alter this code to reflect this. 
 
 

simulated function DrawMouseCursor(Canvas C) 

{ 

   local float XPos, YPos; 

 

   C.SetDrawColor(255, 255, 255); 

   C.Style = ERenderStyle.STY_Alpha; 

 

   // find position of cursor, and clamp it to screen 

   XPos = ExampleController(PlayerOwner).PlayerMouse.X + C.SizeX / 2.0; 

   if (XPos < 0) 

   { 

      ExampleController(PlayerOwner).PlayerMouse.X -= XPos; 

      XPos = 0; 

   } 

   else if (XPos >= C.SizeX) 

   { 

      ExampleController(PlayerOwner).PlayerMouse.X -= (XPos - C.SizeX); 

      XPos = C.SizeX - 1; 

   } 

   YPos = ExampleController(PlayerOwner).PlayerMouse.Y + C.SizeY / 2.0; 

   if (YPos < 0) 

   { 

      ExampleController(PlayerOwner).PlayerMouse.Y -= YPos; 

      YPos = 0; 

   } 

   else if (YPos >= C.SizeY) 

   { 

      ExampleController(PlayerOwner).PlayerMouse.Y -= (YPos - C.SizeY); 

      YPos = C.SizeY - 1; 

   } 

   // render mouse cursor 

   C.SetPos(XPos, YPos); 

   C.DrawTileScaled(MouseCursorTexture, 1.0, 1.0); 

 

   return; 

} 

 

 
Finally we need to decide when we should call this function to display our cursor. 
 
The PostRender function is called after all the other rendering occurs so we can 

be sure that anything drawn here will be on top of all other graphical elements.  
This will ensure that our cursor is always drawn on top. 
 
We only need to drawn the mouse cursor when we’re in the PlayerMousing 

state. 
 
 

simulated event PostRender(Canvas C) 

{ 

   Super.PostRender(C); 

 

   ExampleController(PlayerOwner).LastHUDSizeX = C.SizeX; 

   ExampleController(PlayerOwner).LastHUDSizeY = C.SizeY; 

 

   if (PlayerOwner.IsInState('PlayerMousing')) DrawMouseCursor(C); 

 

   return; 

} 

 

All done 

You should now be able to compile, fix any bugs and run your code.  Switch into 
PlayerMousing state by pressing whatever key you assigned to it.  As you 

move the mouse on screen you should find that the screen coordinates are 
updated.  When you click the mouse you should see Fire appear.  You may 
need to disable the screen co-ordinate display in order to see this – as the Fire 
text will be immediately overwritten. 

Whats next? 

This could easily form the basis for a number of systems within the game.  It is 
simply a case of displaying addition images / text and recording where the player 
clicks on the screen. 

• You could spring up a simply yes / no dialog. 

• You could add the ability to use items in your inventory. 

• You could allow items from your inventory to be combined 
o Adding Drag and Drop support will require considerable more 

thought. 

• You could use it to hold a simple conversation (although there are better 
ways of achieving this) 

• You could use this to have the player predict the murder outcome. 
 
Note: Remember that Unreal scales to run at different resolutions.  Ideally you 
should have any of your on-screen graphical elements scale automatically to the 
screen size.  This isn’t required for the assignment hand-in – you may specify the 
resolution that you wish the game to run in, providing it’s a 4:3 standard (e.g. 
800x600 or 1024x768) 

Next time: 

The next tutorial will show you how to extend this 2D mouse picking into 3D 
scene picking. 

References: 

http://udn.epicgames.com/Two/MouseCursorInterface 
 


